ВИВЧЕННЯ ІНТЕГРАЛА РІМАНА З ВИКОРИСТАННЯМ СКМ MAXIMA

Деканов С. Я.
кандидат фіз.-мат. наук,
Національний педагогічний університет імені М.П.Драгоманова

У роботі розглядається застосування системи комп'ютерної математики Maxima до вивчення теми «Інтеграл Рімана» у педагогічних навчальних закладах.

В работе рассматривается применение системы компьютерной математики Maxima к изучению темы «Интеграл Римана» в педагогических учебных заведениях.

We consider the application of computer mathematics system Maxima to studying of the topic "Riemann integral" in teacher training universities.

1. Вступ. В умовах сьогодення, коли на зміну таблицям Брадіса, логарифмічній лінійці і калькулятору прийшли потужні сучасні системи комп'ютерної математики (СКМ), причому вони вже теж багато років оновлювались і вдосконалювались, оволодіння такими СКМ стало об'єктивною необхідністю для тих, хто вивчає математику та її застосування. Тому природно поєднувати вивчення математичних дисциплін у внз з вивченням сучасних багатофункціональних СКМ, до яких належать: Mathematica, MathLab, Maple, MathCAD, Maxima, Derive та інші. Перелічені СКМ дозволяють розв'язувати математичні задачі як чисельно, так і в символьному поданні, вміщують у собі широку базу математичних знань, мають досконалі засоби графічної візуалізації.

У даній роботі робиться спроба показати, яким чином процес вивчення інтеграла Рімана функцій однієї змінної можна зробити ефективнішим і цікавішим завдяки використанню програми Maxima [1, 2].

За можливостями використання і характеристиками Maxima близька до найбільш потужних професійних систем комп'ютерної математики Mathematica і Maple, які є дорогими комерційними продуктами, тоді як Maxima розповсюджується через Інтернет безкоштовно [3].

2. Короткий опис програми Maxima. Робота з цією програмою здійснюється через певну графічну оболонку, причому найзручніше користуватися графічною оболонкою wxMaxima 0.7.4 з українським інтерфейсом (рис. 1).

Вікно програми wxMaxima містить рядок меню, кнопки редагування, вікно виведення результатів, рядок введення команд і внизу додаткові кнопки для роботи з виразами. У вікні виведення результатів кожен вираз, який вводиться, позначається (%iN), а відповідний йому результат позначається (%oN) (від англ. ***input*** – вхідні дані, ***output*** – результат), де N – номер формули. Окремо взятий символ % набуває значення останнього виведеного виразу. Команди вводяться у рядку введення і починають автоматично опрацьовуватися після натиснення клавіші **Enter.**

Для обчислення за допомогою програми Maxima невизначеного інтеграла $\int f(x)dx$

або R-інтеграла $\int_a^b f(x)dx$ використовуються відповідно команди

integrate(f(x),x)   та   integrate(f(x),x,a,b).

Обчислимо невизначений інтеграл $\int (2z - \sqrt[3]{z} + 3\cos z)dz$.

Для цього у командному рядку вводимо integrate(2*z-z^(1/3)+3*cos(z),z) і, натиснувши **Enter,** дістаємо (рис. 1):



Рис. 1.

Надалі при демонструванні роботи даної програми будемо наводити лише вміст вікна виведення результатів. При цьому команди можна задавати і з командного рядка, і через меню (можна вибирати, що зручніше).

3. Використання Maxima для з'ясування суті поняття інтеграла Рімана та його геометричного змісту. Як відомо [4, с. 194], інтеграл Рімана, або R-інтеграл, функції f на відрізку $[a;b]$ вводиться як границя інтегральної суми: $\int_a^b f(x)dx = \lim_{\lambda(T)\to 0} \sum_{k=0}^{n-1} f(x_k^*)\Delta x_k$. У зв'язку з цим при вивченні цього поняття СКМ можна використати для: 1) обчислення інтегральних сум, 2) ілюстрування суті поняття інтеграла як границі інтегральних сум, 3) обчислення інтеграла у символьному або чисельному вигляді, 4) ілюстрування геометричного змісту інтеграла. Проілюструємо можливі застосування програми Maxima на прикладах розв'язування деяких типів задач.

Задача 1. *Для заданої функції* $f(x)$ *і довільного заданого розбиття* T *відрізка* $[a;b]$ *обчислити інтегральну суму* $S(T,X^*)$ *з машинною точністю. За проміжні точки узяти ліві (праві) кінці відрізків розбиття.*

□ Спершу задаємо функцію, наприклад, $f(x) = e^{-x^2}$.

(%i1) f(x):=exp(-x^2)$

Розбиття T задаємо у вигляді упорядкованого списку. Зауважимо, що цим самим автоматично задається і відрізок $[a;b]$.

(%i2) T:[-2,-1.7,-1.3,-1,-0.5,0,0.25,0.5,0.8,1,1.2,1.6,1.8,2]$

Підраховуємо, скільки елементарних відрізків у даному розбитті:

(%i3) N:length(T)-1$

Складемо інтегральну суму $S(T,X^*)=\sum_{k=1}^{n} f(x_k)\Delta x_k$, узявши ліві кінці в якості проміжних точок. У комп'ютерному поданні $x_k = T[k]$ – це k-тий елемент списку T, причому нумерація повинна починатися від $k=1$.

(%i4) S:sum(f(T[k])*(T[k+1]-T[k]),k,1,N)$

Обчислюємо дану інтегральну суму S:

(%i5) float(S);

(%o5) 1.672013850279529

Для того щоб проміжними точками слугували праві кінці, у рядку (%i4) потрібно було б записати f(T[k+1]). ■

Задача 2. *Для заданої функції* $f(x)$ *і заданого розбиття* T *обчислити верхню й нижню суми Дарбу* $S^*(T)$ *і* $S_*(T)$.

□ Порівняно з попередньою, ця задача набагато складніша, тому що при її розв'язуванні потрібно шукати інфімуми та супремуми функції на відрізках розбиття. У комплект з Maxima включено пакет riemsum.mac, призначений для чисельного обчислення сум Дарбу функції $f(x)$, яка є раціональною функцією, зокрема многочленом. Після завантаження цього пакету стає доступною команда upper_and_lower_sums(f(x),x,T), де T – розбиття відрізка $[a;b]$ у вигляді списку, не обов'язково упорядкованого. В якості результату ця команда повертає список з трьох чисел: $[S^*(T), S_*(T), S^*(T)-S_*(T)]$.

Продемонструємо відшукання сум Дарбу за допомогою пакету riemsum на прикладі функції $f(x) := \dfrac{x^3-2x-1}{x^2+4}$.

(%i1) load(riemsum)$

(%i2) f(x):=(x^3-2*x-1)/(x^2+4)$

Розбиття візьмемо з розв'язання попередньої задачі:

(%i3) T:[-2,-1.7,-1.3,-1,-0.5,0,0.25,0.5,0.8,1,1.2,1.6,1.8,2]$

Знайти суми Дарбу й різницю між ними для заданої функції і заданого розбиття можна за допомогою команди:

(%i4) upper_and_lower_sums(f(x),x,T);

Дістанемо відповідь:

(%o4) [–0.48156168761106, –1.10581480373927, 0.62425311612821]

Точність цих значень не більша за 10^{-7}, оскільки саме з такою точністю запрограмоване відшукання коренів многочленів у пакеті riemsum.mac.

Вручну вводити розбиття T довго і незручно. Для випадку розбиття відрізка $[a;b]$ на n рівних частин у пакеті riemsum.mac є команда make_partition(a,b,n), яка виконує таке розбиття і повертає його точки у вигляді списку. Наприклад, продовжуючи відкриту сесію

Maxima, утворимо розбиття відрізка $[-2;2]$ на 50 частин:

(%i5) T:make_partition(-2,2,50)$

Обчислимо тепер суми Дарбу для цього розбиття:

(%i6) upper_and_lower_sums(f(x),x,T);

(%o6) [–0.70720147773207, –0.86352100977965, 0.15631953204758]

Помічаємо закономірність, що при подрібненні розбиття різниця між верхньою й нижньою сумами Дарбу все ближче наближається до нуля. А ще, оскільки інтеграл Рімана міститься між сумами Дарбу, то за його наближене значення можна прийняти середину між сумами Дарбу:

(%i7) (%[1]+%[2])/2;

(%o7) -0.78536124375586

Отже, $\int_{-2}^{2} \frac{x^3-2x-1}{x^2+4}dx \approx -0.78536$. Зауважимо, що точне значення цього інтеграла дорівнює $-\frac{\pi}{4} \approx -0.78539816339745$. +

Задача 3. *Для заданої функції $f(x) \geq 0$ та заданого відрізка $[a;b]$ обчислити інтегральну суму, яка відповідає розбиттю відрізка на N рівних частин і конкретному способу вибору проміжних точок (наприклад, як середин елементарних відрізків). Проілюструвати геометричний зміст інтегральної суми.*

□ Розглянемо розв'язання цієї задачі на прикладі функції $f(x) = \cos x$, заданої на відрізку $[a;b] = [-\frac{\pi}{2};\frac{\pi}{4}]$, і коли $N = 20$.

Спочатку задамо функцію, відрізок і кількість точок розбиття.

(%i1) f(x):=cos(x)$ a:-%pi/2$ b:%pi/4$ N:20$

Визначимо послідовність розбиттів $T_n = \{x(k,n)\}$ і відповідну послідовність наборів $X_n^* = \{c(k,n)\}$ проміжних точок:

(%i2) x(k,n):=a+k*(b-a)/n$ c(k,n):=(x(k,n)+x(k+1,n))/2$

Визначимо послідовність інтегральних сум $S(n)$:

(%i3) S(n):=sum(f(c(k,n))*(b-a)/n,k,0,n-1)$

Знайдемо точне значення інтеграла:

(%i4) I:integrate(f(x),x,a,b)$

Виведемо на екран значення інтегральної суми при $n = N = 20$:

(%i5) print("S(",N,")=",float(S(N)))$

(%o5) $S(\ 20\ ) = 1.708094395922389$

Для уявлення про ступінь наближення інтегральної суми до інтеграла виведемо ще й точне значення інтеграла.

(%i6) print("I=",I,"=",float(I))$

(%o6) $I = 1/\sqrt{2}+1 = 1.707106781186548$

Аналітичну частину завершено. Перейдемо до графічного зображення. Скористаємося допоміжним пакетом draw.lisp, який служить інтерфейсом між Maxima і графічним редактором gnuplot. Перевагами цього пакету є зручне керування опціями зображень; можливість побудови графіків неявних функцій, многокутників, прямокутників, еліпсів та векторів; зручна можливість заповнення кольором фігур.

Основною командою пакету draw для зображення плоских кривих і фігур є draw2d(параметри, фігури). Скористаємося цією командою для зображення геометричного змісту інтегральної суми. Задамо графічні об'єкти, які треба намалювати. Спочатку визначимо графік функції $f(x)$ на відрізку $[a-0,2;b+0,2]$ з товщиною лінії 2 тч:

(%i7) graph:[line_width=2,explicit(f(x),x,a-0.2,b+0.2)]$

Далі опишемо серію прямокутників, поставлених на відрізки розбиття:

(%i8) pp(n):=makelist(rectangle([x(k,n),0],[x(k+1,n),f(c(k,n))]),k,0,n-1)$

Визначимо параметри малюнка, в даному випадку – колір заповнення фігур (прямокутників pp(n)) і пропорційні осі, тобто однакові масштаби по осях.

(%i9) options:[fill_color=light-blue, proportional_axes=xy]$

Визначимо тепер весь малюнок як один об'єкт. Він буде списком, який складається з фігур та параметрів зображення.

(%i10) mal(n):=append(options,pp(n),graph)$

(Команда append об'єднує декілька списків в один.) Завантажимо пакет draw:

(%i11) load(draw)$

і виконаємо побудову нашого зображення шляхом застосування команди draw2d до об'єкта mal(n) при $n = N = 20$:

(%i12) apply(draw2d,mal(N))$

Після цього з'явиться вікно графічного редактора gnuplot, а в ньому малюнок, який ілюструє геометричний зміст інтегральної суми (рис. 2). ■



Рис. 2.

Задача 4. *Обчислити R-інтеграл заданої функції на заданому відрізку як границю певної послідовності інтегральних сум.*

 Послідовність (T_n) розбиттів відрізка $[a;b]$ природно отримується шляхом його поділу на n рівних частин. А природним способом вибору проміжних точок є вибір лівих, правих чи середніх точок відрізків розбиття. Саме такі послідовності було побудовано при розв'язуванні задачі 3. Тому можна переписати звідти перші три рядки вводу:

(%i1) f(x):=x^2$ a:-1$ b:2$

x(k,n):=a+k*(b-a)/n$

c(k,n):=(x(k,n)+x(k+1,n))/2$

S(n):=sum(f(c(k,n))*(b-a)/n,k,0,n-1)$

Тепер залишається знайти $\lim_{n\to\infty} S(n)$. Але за допомогою команди limit(S(n),n,inf), призначеної для обчислення границь, одразу не вийде обчислити границю такого складного об'єкта, як сума. Тому перш ніж обчислювати границю послідовності $S(n)$ потрібно цю послідовність спростити. Стандартною опцією у Maxima, яка виконує спрощення сум, є опція simpsum, яку потрібно вводити через кóму після суми. Застосуємо опцію simpsum до нашої суми $S(n)$, причому для одержання менш громіздкої відповіді поєднаємо її з опцією ratsimp:

(%i7) S(n),simpsum,ratsimp;

(%o7) $\dfrac{12n^2-9}{4n^2}$

Після такого спрощення знаходження границі стає елементарним:

(%i8) limit(%,n,inf);

(%o8) 3

Таким же чином Maxima допоможе знайти границю інтегральної суми і у випадку, коли $f(x)$ є довільним многочленом. Наприклад, знайдемо інтеграл $\int_{-2}^{3}(x^3-x-1)dx$ як границю інтегральної суми. Для цього повернемося до команди (%i1), перезадамо в її першому рядку функцію і відрізок:

(%i1) f(x):=x^3-x-1$ a:-2$ b:3$

x(k,n):=a+k*(b-a)/n$ c(k,n):=(x(k,n)+x(k+1,n))/2$

S(n):=sum(f(c(k,n))*(b-a)/n,k,0,n-1)$

і переобчислимо цю команду. У результаті буде визначено нову послідовність $S(n)$ інтегральних сум. Знайдемо її границю командою

(%i16) limit(ev(S(n),simpsum),n,inf);

(%o16) $\dfrac{35}{4}$

Можливості опції simpsum, по суті, вичерпуються спрощенням сум вигляду $\sum_k a^k$ і $\sum_{k=m}^{n} p(k)$, де $p(k)$ – многочлен відносно k. Разом з тим, у комплекті з Maxima є додатковий пакет simplify_sum.mac, у якому є значно потужніша за опцію simpsum команда simplify_sum. Вона вміє спрощувати суми, які містять факторіали (чи біноміальні коефіцієнти), а також суми вигляду

$$\sum_{k=m}^{n} p(k)\exp(ak),\ \sum_{k=m}^{n} p(k)\cos(ak),\ \sum_{k=m}^{n} p(k)\sin(ak),$$

де $p(k)$ – многочлен відносно k.

Наведемо декілька прикладів, як працює команда simplify_sum.

(%i17) load(simplify_sum)$

(%i18) sum(k*3^k,k,1,n);

(%o18) $\sum_{k=1}^{n} k3^k$

(%i19) simplify_sum(%);

(%o19) $\frac{(2n-1)3^{n+1}}{4}+\frac{3}{4}$

(%i20) sum(k^2*4^k,k,1,n);

(%o20) $\sum_{k=1}^{n} k^2 4^k$

(%i21) simplify_sum(%);

(%o21) $\frac{\left(9n^2-6n+5\right)4^{n+1}}{27}-\frac{20}{27}$

(%i22) sum(k!/(n+k)!,k,1,n);

(%o22) $\sum_{k=1}^{n} \frac{k!}{(n+k)!}$

(%i23) simplify_sum(%);

(%o23) $\frac{n+1}{(n-1)(n+1)!}-\frac{(n+1)!}{(n-1)(2n)!}$

(%i24) sum(cos(k*x),k,0,n);

(%o24) $\sum_{k=0}^{n} \cos(kx)$

(%i25) simplify_sum(%);

(%o25) $\frac{\frac{e^{i(n+1)x}-1}{e^{ix}-1}+\frac{e^{-i(n+1)x}-1}{e^{-ix}-1}}{2}$

(%i26) trigrat(%)

(%o26) $\frac{\sin\left(\frac{(2n+1)x}{2}\right)+\sin\left(\frac{x}{2}\right)}{2\sin\left(\frac{x}{2}\right)}$

(%i27) sum(k*cos(k*x),k,1,n);

(%o27) $\sum_{k=1}^{n} k\cos(kx)$

(%i28) trigrat(simplify_sum(%));

(%o28) $\dfrac{n\cos((n+1)x)+(-n-1)\cos(nx)+1}{2\cos(x)-2}$

Повернемося до задачі про обчислення границі послідовності інтегральних сум за допомогою системи Maxima. Наведемо ще кілька прикладів, які демонструють можливості даної системи. При цьому розпочнемо новий сеанс роботи.

$$\int_0^1 2^x\,dx = ?$$

(%i1) f(x):=2^x$ a:0$ b:1$

x(k,n):=a+k*(b-a)/n$ c(k,n):=(x(k,n)+x(k+1,n))/2$

S(n):=sum(f(c(k,n))*(b-a)/n,k,0,n-1)$

(%i7) S(n),factor;

(%o7) $\dfrac{\sum\limits_{k=0}^{n-1} 2^{\frac{k}{n}+\frac{1}{2n}}}{n}$

(%i8) %,simpsum;

(%o8) $\dfrac{2^{\frac{1}{2n}}}{n(2^{\frac{1}{n}}-1)}$

(%i9) limit(%,n,inf);

(%o9) $\dfrac{1}{\log(2)}$

$$\int_{-\pi/4}^{3\pi/2} \cos x dx = ?$$

(%i1) f(x):=cos(x)$ a:-%pi/4$ b:3*%pi/2$

x(k,n):=a+k*(b-a)/n$ c(k,n):=(x(k,n)+x(k+1,n))/2$

S(n):=sum(f(c(k,n))*(b-a)/n,k,0,n-1)$

(%i16) S(n),factor;

(%o16) $\dfrac{7\pi\sum\limits_{k=0}^{n-1}\cos\left(\dfrac{\pi(2n-14k-7)}{8n}\right)}{4n}$

(%i17) load(simplify_sum)$

(%i18) simplify_sum(%o16),rectform,trigreduce;

(%o18) $\dfrac{7\sqrt{2}\pi\sin\left(\dfrac{7\pi}{8n}\right)-7\pi\sin\left(\dfrac{7\pi}{8n}\right)}{\left(2^{\frac{5}{2}}\cos\left(\dfrac{7\pi}{4n}\right)-2^{\frac{5}{2}}\right)n}$

(%i19) limit(%,n,inf),expand;

(%o19) $\frac{1}{\sqrt{2}}-1$

Розв’язування задачі 4 підтверджує, що обчислення границь інтегральних сум тісно пов’язане з проблемою спрощення сум, яка є набагато складнішою, ніж обчислення визначених і невизначених інтегралів. ■

Задача 5. *Продемонструвати у середовищі Maxima геометричний зміст інтеграла Рімана функції* $f(x)$, *неперервної і невід’ємної на відрізку* $[a;b]$.

□ Відомо, що інтеграл Рімана $\int_a^b f(x)dx$ неперервної і невід’ємної на відрізку $[a;b]$ функції f виражає собою площу криволінійної трапеції, тобто фігури, обмеженої графіком функції f, віссю OX і прямими $x=a$ та $x=b$.

Покажемо, як зобразити у Maxima заповнену кольором криволінійну трапецію. Це робиться дуже просто за допомогою пакету draw. Наприклад, для зображення криволінійної трапеції, обмеженої графіком функції $x\sin x$, $x\in[0;\pi]$, досить ввести набір команд

```
(%i1) load(draw)$
      draw2d(fill_color=magenta,filled_func=true,
             explicit(x*sin(x),x,0,%pi),
             line_width=5,filled_func=false,
             explicit(x*sin(x),x,0,%pi));
```

У результаті буде побудовано такий рисунок 3:



Рис. 3.

Звичайно, існує ще багато налаштувань зображень, про які можна дізнатися у інтерактивній довідці системи Maxima. ■

4. Використання Maxima для виконання заміни змінної у *R*-інтегралі. У програмі Maxima є вбудована команда changevar(I,x=g(t),t,x), яка дозволяє зробити заміну $x=g(t)$ в інтегралі $I=\int_a^b f(x)dx$, попередньо заданому командою I:’integrate (f(x), x,a,b).

При цьому спочатку на екран буде виведено новий інтеграл по новій змінній та з новими межами інтегрування, а потім його можна буде обчислити командою nouns. Як і при невизначеному інтегруванні, цю формулу доцільно застосовувати, коли треба перевірити ефективність різних підстановок, а також тоді, коли інтеграл не береться безпосередньо командою integrate. Розглянемо декілька прикладів.

Приклад 1. *Обчислити задані інтеграли за допомогою вказаних підстановок:* 1) $\int_0^4 \frac{dx}{1+\sqrt{x}}$, $\sqrt{x}=t$; 2) $\int_0^a \sqrt{\frac{a+x}{a-x}}dx$, $\frac{a+x}{a-x}=t^2$; 3) $\int_0^{\ln 2} \sqrt{e^x-1}dx$, $e^x-1=t^2$; 4) $\int_0^{\pi/2} \frac{dx}{3+2\cos x}$, $t=\operatorname{tg}\frac{x}{2}$.

□ Введемо перший інтеграл:

(%i1) 'integrate(1/(sqrt(x)+1),x,0,4);

(%o1) $\int_0^4 \frac{1}{\sqrt{x}+1}dx$

(%i2) changevar(%o23,sqrt(x)=t,t,x);

Машина запитає:

Is t positive, negative, or zero?

Давши відповідь p, тобто $t>0$, дістаємо

(%o2) $2\int_0^2 \frac{t}{t+1}dt$

Обчислюємо одержаний інтеграл командою

(%i3) %,nouns;

(%o3) $2(2-\log(3))$

До речі, зауважимо, що інтеграл (%o1) командою integrate не береться.

Переходимо до другого інтеграла.

(%i4) 'integrate(sqrt((a+x)/(a-x)),x,0,a/2);

(%o4) $\int_0^{\frac{a}{2}} \sqrt{\frac{x+a}{a-x}}dx$

(%i5) changevar(%o2,(a+x)/(a-x)=t^2,t,x);

(%o5) $-4a\int_{-\sqrt{3}}^{-1} \frac{t|t|}{t^4+2t^2+1}dt$

Для подальшого обчислення цього інтеграла потрібно спростити підінтегральну функцію, розкривши модуль. Судячи з меж інтегрування, робимо припущення

(%i6) assume(t<0)$

Виведемо інтеграл (%o5) на екран ще раз, тепер уже в спрощеному вигляді:

(%i7) %o5,simp;

(%o7) $4a\int\limits_{-\sqrt{3}}^{-1}\frac{t^2}{t^4+2t^2+1}dt$

(%i8) %,nouns;

(%o8) $4\left(\frac{4\pi-3^{\frac{3}{2}}}{24}-\frac{\pi-2}{8}\right)a$

Така сама відповідь виходить і при безпосередньому обчисленні інтеграла (%o4) командою integrate.

Перейдемо до третього інтеграла, очистивши попередньо з пам'яті припущення $t<0$:

(%i9) forget(t<0)$

(%i10) 'integrate(sqrt(%e^x-1),x,0,log(2));

(%o10) $\int\limits_{0}^{\log(2)}\sqrt{e^x-1}dx$

(%i11) changevar(%,%e^x-1=t^2,t,x);

(%o11) $-2\int\limits_{-1}^{0}\frac{t|t|}{t^2+1}dt$

(%i12) assume(t<0)$

(%i13) %o11,simp;

(%o13) $2\int\limits_{-1}^{0}\frac{t^2}{t^2+1}dt$

(%i14) %,nouns;

(%o14) $-\frac{\pi-4}{2}$

Цю відповідь можна перевірити і за допомогою команди integrate, застосованої до інтеграла (%o10).

Очистимо всі факти про змінну t з оперативної пам'яті командою

(%i15) kill(t)$

Нарешті, обчислимо четвертий інтеграл:

(%i16) 'integrate(1/(3+2*cos(x)),x,0,%pi/2);

(%o16) $\int\limits_{0}^{\frac{\pi}{2}}\frac{1}{2\cos(x)+3}dx$

(%i17) changevar(%,tan(x/2)=t,t,x);

(%o17) $2\int\limits_{0}^{1}\frac{1}{\left(2t^2+2\right)\cos\left(2\operatorname{atan}(t)\right)+3t^2+3}dt$

(%i18) %,trigexpand,ratsimp;

(%o18) $2\int_0^1 \frac{1}{t^2+5}dt$

(%i19) %,nouns;

(%o19) $\frac{2\operatorname{atan}\left(\frac{1}{\sqrt{5}}\right)}{\sqrt{5}}$

Зробимо перевірку:

(%i20) %o16,nouns;

(%o20) $\frac{2\operatorname{atan}\left(\frac{1}{\sqrt{5}}\right)}{\sqrt{5}}$

Приклад 1 розв'язано повністю. ■

Приклад 2. *Обчислити задані інтеграли за допомогою вказаних підстановок і проілюструвати ці підстановки графічно:* 1) $\int_1^2 x^2 dx$, $x^2 = t$; 2) $\int_1^2 x^3 dx$, $x^3 = t$; 3) $\int_0^1 \sqrt{4-x^2}\,dx$, $4-x^2=(tx-2)^2$; 4) $\int_0^1 \sqrt{4-x^2}\,dx$, $x = 2\sin t$.

□ Заміну змінної будемо виконувати, як і в попередньому прикладі, але крім цього зображатимемо графіки функцій, що задаються вказаними підстановками.

(%i1) I1:'integrate(x^2,x,1,2)$

(%i2) changevar(I1,x^2=t,t,x);

(%o2) $-\frac{\int_1^4 \sqrt{t}\,dt}{2}$

Очевидно, ця відповідь неправильна, оскільки початковий інтеграл був невід'ємним, а вийшов інтеграл від'ємний. Подивимось графічне зображення зробленої підстановки (рис. 4).

(%i3) load(draw)$

```
draw2d(xaxis=true,yaxis=true,
        xlabel="Ot",ylabel="Ox",
    xaxis_width=2,yaxis_width=2,
    line_width=4,color=blue,
    proportional_axes=xy,xtics=1,ytics=1,
    implicit(x^2=t,t,0,4,x,-2,2));
```



Рис. 4.

Рівняння $x^2 = t$ задає дві функції $x(t)$, а саме $x=-\sqrt{t}$ і $x=\sqrt{t}$ (нижню й верхню вітки параболи). При переході до нової змінної необхідно вибрати якусь одну з цих функцій. Комп'ютер вибирає нижню вітку, бо вона є першим розв'язком при виконанні команди solve(x^2=t,x). А нові межі він знаходить, просто підставляючи старі межі у функцію $t = x^2$.

Перейдемо до інтеграла 2).

(%i5) I2:'integrate(x^3,x,1,2)$

(%i6) changevar(I2,x^3=t,t,x);

(%o6) $$\frac{\left(\sqrt{3}i-1\right)\int_{1}^{8} t^{\frac{1}{3}}dt}{6}$$

Цей результат неправильний, тому що він є уявним комплексним числом. Подивимось по графіку (рис. 5), чи законним було виконання підстановки $x^3 = t$ для даного інтеграла.

(%i7) draw2d(xaxis=true,yaxis=true,xlabel="Ot", ylabel="Ox",
xtics=1, ytics=1,xaxis_width=2, yaxis_width=2,
proportional_axes=xy, line_width=4, color=blue,
implicit(x^3=t,t,-8,8,x,-2,2));



Рис. 5.

З геометричної точки зору все виглядає бездоганно. Лінія $t = x^3$ має рівносильне задання $x = t^{1/3}$, причому для відрізка $[1;2]$ на осі OX очевидно існує відрізок на осі Ot, який відображається на нього. Проте потрібно ще з'ясувати, як комп'ютер шукає функцію $x = x(t)$ з рівності $x^3 = t$. Скоріш за все, він знову бере перший розв'язок, одержаний за допомогою команди:

(%i8) solve(x^3=t,x);

(%o8) $$[x = \frac{\left(\sqrt{3}i-1\right)t^{\frac{1}{3}}}{2}, x = -\frac{\left(\sqrt{3}i+1\right)t^{\frac{1}{3}}}{2}, x = t^{\frac{1}{3}}]$$

Справді, бачимо, що першими йдуть комплексні розв'язки, а потрібний дійсний розв'язок сиротливо притулився в самому кінці. Таким чином, ***при застосуванні команди changevar завжди можуть виникати помилки, коли рівняння $G(t,x)=0$, яке задає підстановку, має комплексні розв'язки відносно x (що можна перевірити командою* solve(G(t,x),x)).**

Так, якщо для обчислення інтеграла 1) замість підстановки $x^2 = t$ зробити підстановку $x = \sqrt{t}$, то дістанемо цілком коректне розв'язання:

(%i9) changevar(I1,x=sqrt(t),t,x);

Is x positive, negative, or zero? p;

(%o9) $\dfrac{\int_1^4 \sqrt{t}\,dt}{2}$

Аналогічно, інтеграл 2) коректно обчислиться підстановкою $x = t^{1/3}$.

Далі розглянемо інтеграл 3) і виконаємо для нього запропоновану другу підстановку Ейлера.

(%i10) I3:'integrate(sqrt(4-x^2),x,0,1)$

(%i11) changevar(I3,4-x^2=(t*x-2)^2,t,x);

(%o11) $-\int_0^{2-\sqrt{3}} \frac{8t^4-16t^2+8}{t^6+3t^4+3t^2+1}dt$

Обчислимо тепер початковий інтеграл і поряд з ним новий інтеграл:

(%i12) [ev(I3,nouns),ev(%o11,nouns)];

(%o12) $[\frac{2\pi+3^{\frac{3}{2}}}{6}, \frac{8\operatorname{atan}\left(\sqrt{3}-2\right)-\sqrt{3}}{2}]$

То наскільки відрізняються обидві відповіді? Переобчислимо їх ще раз, вже чисельно:

(%i13) %o12,numer;

(%o13) $[1.913222954981036, -1.913222954981037]$

Нарешті стало зрозуміло, що новий інтеграл відрізняється від початкового знаком. Побудуємо графік застосованої тут підстановки (рис. 6).

(%i14) draw2d(xaxis=true,yaxis=true,
xlabel="Ot",ylabel="Ox",
xaxis_width=2,yaxis_width=2,
line_width=4,color=blue,
proportional_axes=xy,xtics=2,ytics=1,
implicit(4-x^2=(t*x-2)^2,t,-6,6,x,-3,3));



Рис. 6.

Дослідимо, яку функцію $x(t)$ міг узяти комп'ютер в якості розв'язку рівняння $4 - x^2 = (tx-2)^2$:

(%i15) solve(4-x^2=(t*x-2)^2,x);

(%o15) $[x=\frac{4t}{t^2+1}, x=0]$

Отже, комп'ютер знаходить два розв'язки (що відповідає геометричній ілюстрації), причому для переходу до нової змінної команда changevar могла взяти тільки функцію $x(t)=\frac{4t}{t^2+1}$, і це правильний вибір. Неважко перевірити шляхом підстановки, що й нові межі знайдено правильно.

Чому ж тоді кінцевий результат має помилку в знаці? Звернемо увагу, що команда заміни змінної повертає свій результат у вигляді спрощеного інтеграла. У заданому інтегралі 3) був присутній корінь квадратний, а у відповідному йому інтегралі (%o11) ніяких радикалів уже нема. Це наводить на думку, що неправильний знак може виникати при спрощенні радикалів. Підставимо "вручну" вираз $\frac{4t}{t^2+1}$ в підінтегральну функцію $\sqrt{4-x^2}$ і подивимося, як Maxima спростить отриманий вираз.

(%i16) sqrt(4-x^2),x=(4*t)/(t^2+1);

(%o16) $\sqrt{4-\frac{16t^2}{\left(t^2+1\right)^2}}$

Серед основних команд спрощування (ratsimp, factor, radcan) розкрити корінь може тільки остання команда. Тому застосуємо її для спрощення нашого виразу:

(%i17) radcan(%o16);

(%o17) $\frac{2t^2-2}{t^2+1}$

Легко бачити, що в заданих межах інтегрування, тобто при $t\in[0;2-\sqrt{3}]$, вираз (%o17) від'ємний, і він не годиться для представлення виразу (%o16). Це і є причиною помилкового знаку в інтегралі (%o11).

Отже, ***при спрощенні виразів потрібно обережно користуватися командою radcan. А якщо робиться заміна змінної у визначеному інтегралі командою changevar і при цьому зникають знаки коренів, то дуже вірогідно, що може виникнути помилка.*** На жаль, на результат команди changevar користувач вплинути не може.

Перейдемо до розв'язування пункту 4) прикладу 2. Оскільки інтеграл I3 вже задано, то одразу зробимо в ньому тригонометричну заміну:

(%i18) changevar(I3,x=2*sin(t),t,x);

solve: using arc-trig functions to get a solution.

Some solutions will be lost.

(%o18) $4i\int_0^{\frac{\pi}{6}}\cos(t)\sqrt{\sin(t)-1}\sqrt{\sin(t)+1}dt$

Зобразимо графічну ілюстрацію даної підстановки (рис. 7).

(%i19) draw2d(xaxis=true,yaxis=true,xlabel="Ot",ylabel="Ox",

xaxis_width=2,yaxis_width=2,line_width=4,color=blue,
proportional_axes=xy,xtics=2,ytics=1,explicit(2*sin(t),t,-6,6));



Рис. 7.

З малюнка видно, що дана підстановка підходить для переходу до нової змінної, тільки потрібно правильно вибрати нові межі. Скоріш за все, команда changevar спочатку шукає обернену функцію $t(x)$ до заданої функції $x(t)$, а потім шукає нові межі, як значення функції $t(x)$ при відповідних старих межах. При відшуканні оберненої функції якраз і виникала неоднозначна ситуація, про яку говорилося в супровідному повідомленні.

Неважко пересвідчитися, що межі в новому інтегралі вибрані правильно. А от підінтегральна функція за формою не виглядає прийнятно, бо містить комплексний множник i та від'ємний вираз $\sin t - 1$ під знаком одного з коренів. Можливо, за змістом вивід (%o18) правильно виражає шуканий інтеграл?.. Спростимо його, а потім обчислимо:

(%i20) rootscontract(%o18);

(%o20) $4i\int_0^{\frac{\pi}{6}} \cos(t)\sqrt{\sin(t)^2 - 1}\,dt$

(%i21) trigsimp(%);

(%o21) $-4\int_0^{\frac{\pi}{6}} \cos(t)|\cos(t)|\,dt$

Далі треба підказати комп'ютеру, який знак у виразу під знаком модуля. Враховуючи межі інтегрування, задаємо припущення

(%i22) assume(cos(t)>0)$

(%i23) %o21,simp;

(%o23) $-4\int_0^{\frac{\pi}{6}} \cos(t)^2\,dt$

Обчислимо тепер початковий інтеграл 4) і поряд його значення після заміни змінної:

(%i24) [ev(I3,nouns),ev(%o23,nouns)];

(%o24) $[\frac{2\pi + 3^{\frac{3}{2}}}{6}, -\frac{2\pi + 3^{\frac{3}{2}}}{6}]$

Отже, виявилось, що помилка при заміні змінної в інтегралі все-таки вкралася – неправильний знак мінус. Виокремимо чіткіше причину цієї помилки. До проведення усіляких спрощень інтеграл (%o18) повинен був мати вигляд

(%i25) ’integrate(sqrt(4-4*sin(t)^2)*2*cos(t),t,0,%pi/6);

(%o25) $2\int_{0}^{\frac{\pi}{6}}\cos(t)\sqrt{4-4\sin(t)^2}\,dt$

Оскільки в інтегралі (%o18) один корінь розпався на два, то напевно за алгоритмом команди changevar на підінтенгральну функцію подіяла команда radcan. Справді,

(%i26) radcan(%o25);

(%o26) $4i\int_{0}^{\frac{\pi}{6}}\cos(t)\sqrt{\sin(t)-1}\sqrt{\sin(t)+1}dt$

Бачимо, що підтвердився результат (%o18), що є опосередкованим доказом застосування команди radcan в алгоритмі команди changevar, а в даному випадку її застосовувати було недоцільно. Для того щоб повернутися до початкового вигляду інтеграла, тобто до (%o25), необхідно виконувати обернену операцію rootscontract, яка збирає корені докупи. При цьому, якщо комплексний множник i стоїть безпосередньо біля кореня, то він вноситься під корінь. А якщо цей множник винесений за знак інтеграла, то такого внесення не відбувається. Порівняємо результат (%o20) дії rootscontract на інтеграл (%o18) (де i так і залишилося за знаком інтеграла) і окремо на його підінтегральну функцію:

(%i26) 4*%i*cos(t)*sqrt(sin(t)-1)*sqrt(sin(t)+1);

(%o26) $4i\cos(t)\sqrt{\sin(t)-1}\sqrt{\sin(t)+1}$

(%i27) rootscontract(%);

(%o27) $4\cos(t)\sqrt{1-\sin(t)^2}$

Отже, при розв’язуванні інтеграла 4) прикладу 2 було виявлено ***недоречності в роботі команди changevar, пов’язані з наслідками застосування нею команди radcan, автоматично усунути які певною командою в Maxima неможливо.*** ■

Для усунення виявлених недоліків убудованої в Maxima команди заміни змінної changevar, а також для отримання більшого контролю над процесом заміни змінної пропонуємо визначити нову команду chv(expr,G,t,x). За своїм призначенням вона еквівалентна команді changevar(expr,G,t,x), тільки будується на іншому алгоритмі. При визначенні цієї команди використовуються оператори програмування та деякі нові функції системи Maxima, опис яких можна знайти в довідці Maxima. Доцільно зберегти означення команди chv у файл Maxima-5.21.0\share\maxima\5.21.0\share\integration\changevar1.mac, після чого вона стане доступною через load(changevar1).

```
chv(expr,G,t,x):=(block([g,h,a,b,a1,b1,G1,G2,n1,n2,i0,gpr,ghpr,apr,bpr],
if string(part(expr,0))="integrate" and  part(expr,2)=x
then (f:part(expr,1),
    if length(expr)=2
    then (G1:sort(solve(G,x)),n1:length(G1),i0:0,
        for i:1 step 1 while (i<=n1 and is(scalarp(part(G1[i],2)))) do i0:i0+1,
        if i0<n1
        then (i0:i0+1,
            if part(G1[i0],1)=x and freeof(x,part(G1[i0],2))
```

```
            then (g:part(G1[i0],2),
                 if freeof (%i,g)
                 then 'integrate(subst(g,x,f)*diff(g,t),t)
                 else (print("– Function",x,"(",t, ")","=",g,"is complex!.."),
                 expr))
            else (print("– It is impossible to express",x,"(",t,").")),expr))
        else (print("– All solutions",x,"(",t,")", "are constant!.."),expr))
    else (a:part(expr,3),b:part(expr,4),G1:sort(solve(G,x)),
        n1:length(G1),G2:sort(solve(G,t)),n2:length(G2),
        a1:%gamma,b1:%gamma,g:1,i0:0,gpr:false,ghpr:false,
        for i:1 step 1 while (i<=n1 and is(scalarp(part(G1[i],2)))) do i0:i0+1,
        if i0<n1
        then (i0:i0+1,apr:plus,bpr:minus,
            for i:i0 step 1 unless (i>n1 or ghpr)
              do (if part(G1[i],1)=x and freeof(x,part(G1[i],2))
                     then (g:part(G1[i],2),gpr:true)
                     else g:1,
                   for j:1 step 1 unless (j>n2 or ghpr)
                   do (h:part(G2[j],2),a1:limit(h,x,a,plus),
                      b1:limit(h,x,b,minus),
                      if a1<=b1
                      then (apr:plus,bpr:minus)
                      else (apr:minus,bpr:plus),
                      if string(float(a))=string(float(limit(g,t,a1,apr)))
                      and string(float(b))=string(float(limit(g,t,b1,bpr)))
                      then ghpr:true )),
            if ghpr
            then (if not(freeof(%i,g))
                 then print("– Function",x,"(",t, ")","=",g,"is complex!.."),
                 if imagpart(a1)#0 or imagpart(b1)#0
                 then print("– New bounds are complex!.."),
                 'integrate(subst(g,x,f)*diff(g,t),t,a1,b1))
            else (if not(gpr)
                 then print("– It is impossible to express",x,"(",t,").")
                 else print("– Substitution is not correct!"),expr))
        else (print("– All solutions",x,"(",t,")", "are constant!.."),expr)))
else (if string(part(expr,0))#"integrate"
    then (print("– Expression must be a single integral
              without any coefficients!.."),return(expr)),
    if part(expr,2)#x
    then print("– The 4-th argument must be marked as",part(expr,2),"!.."),expr)
))$
```

Описану вище програму-команду chv(expr,G,t,x) можна застосовувати як до визначених, так і до невизначених інтегралів. Вираз expr повинен являти собою тільки один інтеграл, причому без коефіцієнтів.

Розв’яжемо всі інтеграли прикладу 2 за допомогою нової команди заміни змінної chv.

(%i1) load(changevar1)$

(%i2) ’integrate(x^2,x,1,2);

(%o2) $\int\limits_{1}^{2} x^2\,dx$

(%i3) chv(%o2,x^2=t,t,x);

(%o3) $\dfrac{\int\limits_{1}^{4} \sqrt{t}\,dt}{2}$

(%i4) 'integrate(x^3,x,1,2);

(%o4) $\int\limits_{1}^{2} x^3\,dx$

(%i5) chv(%o4,x^3=t,t,x);

(%o5) $\dfrac{\int\limits_{1}^{8} t^{\frac{1}{3}}\,dt}{3}$

(%i6) 'integrate(sqrt(4-x^2),x,0,1);

(%o6) $\int\limits_{0}^{2-\sqrt{3}} \sqrt{4-\dfrac{16t^2}{(t^2+1)^2}}\left(\dfrac{4}{t^2+1}-\dfrac{8t^2}{(t^2+1)^2}\right)dt$

(%i7) factor(%)

(%o7) $-8\int\limits_{0}^{2-\sqrt{3}} \dfrac{(t^2-1)\sqrt{t^4-2t^2+1}}{(t^2+1)^3}dt$

Зараз необхідно подумати. Під знаком кореня квадратного міститься вираз $t^4-2t^2+1=(t^2-1)^2$, але він не спростився. На жаль, команда factor не заходить вглиб ірраціональних виразів і саме через це повний квадрат залишився непоміченим. Позбутися цього кореня, а відтак, завершити обчислення інтеграла можна двома шляхами: вдумливим використанням команди radcan або вручну замінивши його на потрібний вираз. Пояснимо детальніше обидва способи.

Перший спосіб. Оскільки radcan завжди спрощує таким чином: $\sqrt{x^2-2ax+a^2}=x-a$ (де a – число або якась літера латинського алфавіту, яка йде перед x), то $\sqrt{t^4-2t^2+1}$ буде спрощено до t^2-1. Враховуючи, що $t\in[0;2-\sqrt{3}]$, треба було б той корінь спростити до $1-t^2$. Тому якщо нічого не коригувати і довести розв'язування цим шляхом до кінця, то отриманий результат буде відрізнятися від правильного знаком.

Другий спосіб. Краще зробити так. Виділити вміст рядка (%o7), натиснути по черзі **F5, F4.** Тепер можна редагувати потрібний вираз у рядку введення. Щоб інтеграл поки що не обчислився, перед ним слід поставити апостроф. Корінь $\sqrt{t^4-2t^2+1}$ витираємо і замість нього пишемо $1-t^2$:

(%i8) -8*'integrate(((t^2-1)*(1-t^2))/(t^2+1)^3,t,0,2-sqrt(3));

(%o8) $-8\int_{0}^{2-\sqrt{3}} \frac{(1-t^2)(t^2-1)}{(t^2+1)^3}dt$

(%i9) %,nouns;

(%o9) $-\frac{8\operatorname{atan}\left(\sqrt{3}-2\right)-\sqrt{3}}{2}$

Застосуємо тепер до інтеграла (%o6) тригонометричну підстановку:

(%i10) chv(%o6,x=2*sin(t),t,x);

solve: using arc-trig functions to get a solution.

Some solutions will be lost.

(%o10) $2\int_{0}^{\frac{\pi}{6}} \cos(t)\sqrt{4-4\sin(t)^2}\,dt$

(%i11) trigsimp(%o10);

(%o11) $4\int_{0}^{\frac{\pi}{6}} \cos(t)|\cos(t)|dt$

(%i12) assume(cos(t)>0)$

(%i13) %o11,simp;

(%o13) $4\int_{0}^{\frac{\pi}{6}} \cos(t)^2\,dt$

(%i14) %,nouns;

(%o14) $\frac{2\pi+3^{\frac{3}{2}}}{6}$

Таким чином, на відміну від штатної команди changevar, нова команда chv дозволила безпомилково розв'язати усі завдання прикладу 2. Розглянемо ще одну збірку цікавих інтегралів.

Приклад 3. *Обчислити наступні інтеграли методом підстановки:*

1) $\int_{2}^{4}\frac{\sqrt{x^2-4}}{x^4}dx$; 2) $\int_{0}^{\frac{\pi}{4}}\frac{dx}{a^2\cos^2 x+b^2\sin^2 x}$ $(a>0,b>0)$; 3) $\int_{-a}^{a}\sqrt[3]{\frac{x-2a}{x+2a}}dx$ $(a>0)$;

4) $\int_{0}^{1}\frac{\sqrt{x}}{\sqrt{x}+1}dx$; 5) $\int_{\pi/2}^{3\pi/2}\sqrt{\cos x}\sin(x)dx$.

□ 1) Щоб позбутися кореня, використаємо формулу $\operatorname{tg}^2 t=\frac{1}{\cos^2 t}-1$ і зробимо заміну $x = \dfrac{2}{\cos t} = 2\sec t$:

(%i1) load(changevar1)$

(%i2) 'integrate(sqrt(x^2-4)/x^4,x,2,4);

(%o2) $\int\limits_{2}^{4} \frac{\sqrt{x^2-4}}{x^4} dx$

(%i3) chv(%o2,x=2*sec(t),t,x);

solve: using arc-trig functions to get a solution.

Some solutions will be lost.

(%o3) $\dfrac{\int\limits_{0}^{\frac{\pi}{3}} \frac{\sqrt{4\sec(t)^2-4}\tan(t)}{\sec(t)^3} dt}{8}$

(%i4) trigsimp(%o3);

(%o4) $\dfrac{\int\limits_{0}^{\frac{\pi}{3}} \sin(t)|\sqrt{\cos(t)}||\sin(t)|\,dt}{4}$

(%i5) apply(assume,[cos(t)>0,sin(t)>0]);

(%o5) $[\cos(t) > 0, \sin(t) > 0]$

(%i6) %o4,simp;

(%o6) $\dfrac{\int\limits_{0}^{\frac{\pi}{3}} \cos(t)\sin(t)^2\,dt}{4}$

(%i7) %,nouns;

(%o7) $\dfrac{\sqrt{3}}{32}$

Спробуємо ще застосувати до інтеграла 1) третю підстановку Чебишова:

(%i8) chv(%o2,x^2-4=t^2*x^2,t,x);

– Function x (t) = $\dfrac{2i}{\sqrt{t^2-1}}$ is complex!..

(%o8) $\dfrac{i\int\limits_{-\frac{\sqrt{3}}{2}}^{0} t\sqrt{t^2-1}\sqrt{-\frac{4}{t^2-1}-4}\,dt}{8}$

Попередження про комплексну функцію не означає помилку, адже судячи за межами інтегрування $|t|<1$ і тому вираз $\frac{2i}{\sqrt{t^2-1}}=\frac{2}{\sqrt{1-t^2}}$ є комплексним тільки за формою, а за змістом – дійсним. Тому продовжимо розв'язування:

(%i9) factor(%o8);

(%o9) $$\frac{\int\limits_{-\frac{\sqrt{3}}{2}}^{0} t^2\,dt}{4}$$

(%i10) %,nouns;

(%o10) $$\frac{\sqrt{3}}{32}$$

Після спрощення стало зрозуміло, наскільки ефективною є підстановка Чебишова для даного інтеграла і комп'ютер виконав її правильно.

Обчислимо ще інтеграл 1) за допомогою першої підстановки Ейлера.

(%i11) chv(%o2,x^2-4=(x-t)^2,t,x);

(%o11) $$-16\int\limits_{4-2\sqrt{3}}^{2}\frac{t^4\left(1-\frac{t^2+4}{2t^2}\right)\sqrt{\frac{\left(t^2+4\right)^2}{4t^2}-4}}{\left(t^2+4\right)^4}dt$$

(%i12) factor(%);

(%o12) $$-4\int\limits_{-2(\sqrt{3}-2)}^{2}\frac{(t^2-4)\sqrt{t^4-8t^2+16}|t|}{(t^2+4)^4}dt$$

(%i13) assume(t>0)$

(%i14) %o12,simp;

(%o14) $$-4\int\limits_{-2(\sqrt{3}-2)}^{2}\frac{t(t^2-4)\sqrt{t^4-8t^2+16}}{(t^2+4)^4}dt$$

Знайдемо приблизне значення нижньої межі:

(%i15) -2*(sqrt(3)-2),numer;

(%o15) 0.53589838486225

Корінь $\sqrt{t^4-8t^2+16}$ в інтегралі (%o14) розкриємо вручну, замінивши його на $(4-t^2)$, враховуючи, що $|t|\le 2$. Коефіцієнт -4 перед інтегралом поки що приберемо (щоб мати можливість одразу зробити ще одну заміну змінної в отриманому інтегралі), але потім треба не забути на нього домножити.

(%i16) 'integrate((t*(t^2-4)*(4-t^2))/(t^2+4)^4,t,-2*(sqrt(3)-2),2);

(%o16) $$\int\limits_{-2(\sqrt{3}-2)}^{2}\frac{t(4-t^2)(t^2-4)}{(t^2+4)^4}dt$$

(%i17) chv(%o16,t^2=u,u,t);

(%o17) $$\frac{\displaystyle\int_{28-16\sqrt{3}}^{4} \frac{(4-u)(u-4)}{(u+4)^4} du}{2}$$

(%i18) (-4)*%o17,nouns,ratsimp;

(%o18) $$\frac{\sqrt{3}}{32}$$

Отже, на прикладі одного інтеграла 1) за допомогою комп'ютера було виявлено, яка з рекомендованих у теорії підстановок дозволяє звести його до найпростішого вигляду.

Перед розв'язуванням кожного наступного інтеграла прикладу 3 перезапускатимемо сеанс Maxima.

2) Цей інтеграл цікавий наявністю параметрів a, b. Теорія рекомендує робити для нього заміну $t = \operatorname{tg} x$, хоч можна спробувати і такі підстановки, як $t = \operatorname{ctg} x$ або $t = \operatorname{tg}\frac{x}{2}$.

Зупинимо вибір на першій підстановці.

(%i1) load(changevar1)$

(%i2) 'integrate(1/(a^2*cos(x)^2+b^2*sin(x)^2),x,0,%pi/4);

(%o2) $$\int_0^{\frac{\pi}{4}} \frac{1}{b^2\sin(x)^2 + a^2\cos(x)^2} dx$$

(%i3) chv(%o2,t=tan(x),t,x);

solve: using arc-trig functions to get a solution.

Some solutions will be lost.

(%o3) $$\int_0^1 \frac{1}{(t^2+1)(\frac{b^2t^2}{t^2+1} + \frac{a^2}{t^2+1})} dt$$

(%i4) ratsimp(%);

(%o4) $$\int_0^1 \frac{1}{b^2t^2 + a^2} dt$$

(%i5) %,nouns;

(%o5) $$\frac{\operatorname{atan}\left(\frac{b}{a}\right)}{ab}$$

Інтеграл 2) обчислено.

3) Даний інтеграл цікавий тим, що він містить параметр і в підінтегральній функції, і в межах інтегрування. Крім того, в середовищі Maxima безпосереднім інтегруванням він не обчислюється.

(%i1) load(changevar1)$

(%i2) 'integrate(((x-2*a)/(x+2*a))^(1/3),x,-a,a);

(%o2) $$\int_{-a}^{a} \frac{(x-2a)^{\frac{1}{3}}}{(x+2a)^{\frac{1}{3}}} dx$$

(%i3) chv(%o2,(x-2*a)/(x+2*a)=t^3,t,x);

(%o3) $$\int_{-3^{\frac{1}{3}}}^{-\frac{1}{3^{\frac{1}{3}}}} \frac{\left(-\frac{2at^3+2a}{t^3-1}-2a\right)^{\frac{1}{3}}\left(\frac{3t^2(2at^3+2a)}{(t^3-1)^2}-\frac{6at^2}{t^3-1}\right)}{\left(2a-\frac{2at^3+2a}{t^3-1}\right)^{\frac{1}{3}}} dt$$

(%i4) ratsimp(%);

(%o4) $$12a\int_{-3^{\frac{1}{3}}}^{-\frac{1}{3^{\frac{1}{3}}}} \frac{t^3}{t^6-2t^3+1} dt$$

(%i5) %,nouns,logcontract;

(%o5) $$\frac{\left(4\sqrt{3}\operatorname{atan}\left(\frac{\sqrt{3}-2\,3^{\frac{5}{6}}}{3}\right)-4\sqrt{3}\operatorname{atan}\left(\frac{\sqrt{3}-2\,3^{\frac{1}{6}}}{3}\right)-3^{\frac{5}{3}}+3^{\frac{4}{3}}\right)a}{3}$$

Інтеграл 3) обчислено точно і красиво, у символьному вигляді. А чи можливо дістати якесь підтвердження правильності результату (%o5)?

Для цього можна скористатися іншим засобом комп'ютерної математики, наприклад, програмою Gran1.

Переобчислимо чисельно вираз (%o5):

(%i6) %o5,numer;

(%o6) $-2.041614878242466a$

До речі, штатна команда

(%i7) changevar(%o5,(x-2*a)/(x+2*a)=t^3,t,x);

приводить до інтеграла з комплексними межами:

(%o7) $$12a\int_{-\frac{3^{\frac{5}{6}}i-3^{\frac{1}{3}}}{2}}^{-\frac{\sqrt{3}i-1}{2\,3^{\frac{1}{3}}}} \frac{t^3}{t^6-2t^3+1} dt$$

Зрозуміло, що за допомогою цього інтеграла коректно обчислити інтеграл 3) неможливо.

4) Даний інтеграл цікавий тим, що безпосереднім інтегруванням у програмі Maxima він не береться:

(%i1) integrate(sqrt(x)/(sqrt(x)+1),x,0,1);

(%o1) $\int_0^1 \frac{\sqrt{x}}{\sqrt{x}+1}dx$

Проте відповідний невизначений інтеграл береться:

(%i2) integrate(sqrt(x)/(sqrt(x)+1),x);

(%o2) $-4(\sqrt{x}+1)+(\sqrt{x}+1)^2+2\log(\sqrt{x}+1)$

Маючи первісну, досить просто скористатися формулою Ньютона – Лейбніца:

(%i3) ev(%,x=1)-ev(%,x=0);

(%o3) $2\log(2)-1$

Другим способом обчислення інтеграла 4) є, звичайно, спосіб заміни змінної:

(%i4) changevar(%o1,t=sqrt(x),t,x);

Is t positive, negative, or zero?p;

(%o4) $2\int_0^1 \frac{t^2}{t+1}dt$

(%i5) %,nouns;

(%o5) $2\log(2)-1$

5) Спробуємо обчислити даний інтеграл безпосередньо:

(%i1) ’integrate(sqrt(cos(x))*sin(x),x,%pi/2,3*%pi/2);

(%o1) $\int_{\frac{\pi}{2}}^{\frac{3\pi}{2}} \sqrt{\cos(x)}\sin(x)dx$

(%i2) %,nouns;

(%o2) 0

Тепер спробуємо обчислити інтеграл 5) за допомогою вбудованої команди заміни змінної:

(%i3) changevar(%o1,cos(x)=t,t,x);

(%o3) 0

Двома способами Maxima видала однаковий результат.

Зауважимо, що у прикладі 3, 5) підінтегральна функція як дійсна функція не визначена в жодній точці всередині проміжку інтегрування. Проте ця функція визначена і неперервна на заданому відрізку як комплекснозначна функція дійсної змінної. Система Maxima дозволяє обчислювати інтеграли і від таких функцій. Так, якщо $a<0$, то в якості кореня $\sqrt{a}$ береться його головне значення, а саме $i\sqrt{-a}$. Тому й виходить, що

$$\int_{\pi/2}^{3\pi/2} \sqrt{\cos x}\sin x dx = i\int_{\pi/2}^{3\pi/2} \sqrt{-\cos x}\, d(-\cos x) = i\int_0^0 \sqrt{t}\,dt = 0.$$

Взагалі, при інтегруванні функції $x^{\frac{1}{n}} = \sqrt[n]{x}$ по відрізку $[a;b]\subset(-\infty;0]$ у випадку непарного n в якості значень $\sqrt[n]{x}$ беруться дійсні значення, а у випадку парного n – головні значення: $\sqrt[n]{x}=\sqrt[n]{|x|}(\cos\frac{\arg x}{n}+i\sin\frac{\arg x}{n})$. Наприклад,

(%i4) integrate(x^(1/5),x,-32,0);

(%o4) $-\frac{160}{3}$

(%i5) integrate(x^(1/4),x,-16,0);

(%o5) $\frac{128(-1)^{\frac{1}{4}}}{5}$

(%i6) rectform(%);

(%o6) $\frac{2^{\frac{13}{2}}i}{5}+\frac{2^{\frac{13}{2}}}{5}$

Цю особливість системи Maxima слід також мати на увазі під час обчислення інтегралів. ■

5. Використання Maxima для обчислення *R*-інтегралів методом інтегрування частинами. У комплекті з даною програмою є додатковий пакет bypart.mac, який містить команду відшукання невизначеного інтеграла методом інтегрування частинами:

byparts(exp,x,u,dv):=(dv:integrate(dv,x),u*dv-integrate(dv*diff(u,x,1),x));

Оскільки аналогічної команди для *R*-інтеграла немає, то можна визначити її самостійно, додавши у файл bypart1.mac такий рядок:

byparts2(x,u,dv,a,b):=(block([v],v:integrate(dv,x),subst(b,x,u*v)-
subst(a,x,u*v)-’integrate(v*rat(diff(u,x)),x,a,b)))$

В аргументах цієї команди задається ліва частина формули інтегрування частинами (змінна інтегрування, вирази, які ми хочемо позначити за u, v', та межі інтегрування). Результатом виконання наведеної вище команди є права частина формули інтегрування частинами (причому вираз $uv\big|_a^b$ буде одразу обчислено, а інтеграл $\int_a^b v\,du$ буде виведено у необчисленій формі, щоб можна було його проаналізувати).

Приклад 4. *Обчислити наступні інтеграли:* 1) $\int_{-1}^{1} x2^x\,dx$; 2) $\int_0^1 \ln(1+x^2)dx$; 3) $\int_{\pi/4}^{\pi/3}\frac{xdx}{\sin^2 x}$; 4) $\int_0^{1/2}\arcsin^2 x dx$; 5) $\int_0^{\pi/2}\sin 2x \operatorname{arctg}(\sin x)dx$.

□ Інтеграли 1) – 3) обчислюються методом інтегрування частинами звичайно. Тому просто проілюструємо обчислення їх у системі Maxima.

(%i1) load(bypart1)$

(%i2) byparts2(x,x,2^x,-1,1);

(%o2) $\dfrac{5}{2\log(2)}-\dfrac{\int_{-1}^{1} 2^{x}\,dx}{\log(2)}$

(%i3) %,nouns;

(%o3) $\dfrac{5}{2\log(2)}-\dfrac{3}{2\log(2)^{2}}$

(%i4) byparts2(x,log(1+x^2),1,0,1);

(%o4) $\log(2)-2\int_{0}^{1}\dfrac{x^{2}}{x^{2}+1}dx$

(%i5) %,nouns;

(%o5) $\log(2)+\dfrac{\pi-4}{2}$

(%i6) byparts2(x,x,1/sin(x)^2,%pi/4,%pi/3);

(%o6) $\int_{\frac{\pi}{4}}^{\frac{\pi}{3}}\dfrac{1}{\tan(x)}dx-\dfrac{\pi}{3^{\frac{3}{2}}}+\dfrac{\pi}{4}$

(%i7) %,nouns;

(%o7) $\log\left(\dfrac{\sqrt{3}}{2}\right)+\dfrac{\log(2)}{2}-\dfrac{\pi}{3^{\frac{3}{2}}}+\dfrac{\pi}{4}$

4) Для обчислення даного інтеграла потрібно двічі застосувати формулу інтегрування частинами.

(%i8) byparts2(x,asin(x)^2,1,0,1/2);

(%o8) $\dfrac{\pi^{2}}{72}-2\int_{0}^{\frac{1}{2}}\dfrac{x\operatorname{asin}(x)}{\sqrt{1-x^{2}}}dx$

(%i9) byparts2(x,asin(x),-2*x/sqrt(1-x^2),0,1/2);

(%o9) $\dfrac{\pi}{2\sqrt{3}}-1$

(%i10) %pi^2/72+%;

(%o10) $\dfrac{\pi^{2}}{72}+\dfrac{\pi}{2\sqrt{3}}-1$

5) Обчислення цього інтеграла доцільно розпочати із заміни змінної $\sin x=t$, а вже потім продовжити інтегрувати частинами.

(%i11) load(changevar1)$

(%i12) ’integrate(sin(2*x)*atan(sin(x)),x,0,%pi/2);

(%o12) $\int_0^{\frac{\pi}{2}} \sin(2x)\operatorname{atan}(\sin(x))dx$

(%i13) chv(%o12,sin(x)=t,t,x);

solve: using arc-trig functions to get a solution.

Some solutions will be lost.

(%o13) $\int_0^1 \frac{\operatorname{atan}(t)\sin(2\operatorname{asin}(t))}{\sqrt{1-t^2}}dt$

(%i14) trigexpand(%);

(%o14) $2\int_0^1 t\operatorname{atan}(t)dt$

(%i15) byparts2(t,atan(t),2*t,0,1);

(%o15) $\frac{\pi}{4} - \int_0^1 \frac{t^2}{t^2+1}dt$

(%i16) %,nouns,expand;

(%o16) $\frac{\pi}{2} - 1$

Знайдемо наближене десяткове значення цього інтеграла:

(%i17) %o16,numer;

(%o17) 0.5707963267949

А тепер виконаємо безпосереднє обчислення інтеграла 5):

(%i18) %o12,nouns;

(%o18) $-\frac{\pi}{8} - 1$

Яка прикра помилка! До цього команда integrate не робила таких непояснюваних помилок і викликала повагу своєю потужністю й коректністю. Це ще раз підтверджує тезу про те, що далеко не всі результати, отримані за допомогою систем комп'ютерної математики, можна вважати доконаними фактами. ■

Метод інтегрування частинами часто застосовують до виведення рекурентних формул для обчислення інтегралів.

Приклад 5. *Обчислити інтеграл* $I = \int_0^{\pi/b} e^{ax}\sin bx dx \ \ (a > 0, b > 0)$.

□ Застосуємо формулу інтегрування частинами:

(%i1) load(bypart1)$

(%i2) byparts2(x,%e^(a*x),sin(b*x),0,%pi/b);

(%o2) $\frac{a\int_0^{\frac{\pi}{b}} e^{ax}\cos(bx)dx}{b}+\frac{e^{\frac{\pi a}{b}}}{b}+\frac{1}{b}$

Далі застосуємо формулу інтегрування частинами до інтеграла, який утворився, причому за u знову позначимо e^{ax}:

(%i3) byparts2(x,%e^(a*x),cos(b*x),0,%pi/b);

(%o3) $-\frac{a\int_0^{\frac{\pi}{b}} e^{ax}\sin(bx)dx}{b}$

Отже, $I=\%o2$, де інтеграл в %o2 дорівнює $\%o3=-\frac{a}{b}I$. Складемо рівняння:

(%i4) (a*(-a)/b*I)/b+%e^((%pi*a)/b)/b+1/b=I;

(%o4) $-\frac{a^2 I}{b^2}+\frac{e^{\frac{\pi a}{b}}}{b}+\frac{1}{b}=I$

Розв'язавши це рівняння, тим самим знайдемо шуканий інтеграл:

(%i5) solve(%,I),factor;

(%o5) $[I=\frac{b\left(e^{\frac{\pi a}{b}}+1\right)}{b^2+a^2}]$ ■

6. Висновки. Вище розглянуто можливості застосування системи Maxima при вивченні поняття *R*-інтеграла, причому і в якості методичного засобу для супроводу навчального процесу, і в якості серйозного математичного інструменту, який повинен допомагати виконувати точні обчислення *R*-інтегралів.

Подаючи навіть нескладні інтеграли у вигляді границі інтегральних сум, ми зіткнулися зі складною проблемою спрощування суматорних виразів, у розв'язанні якої Maxima може суттєво допомогти, проте далеко не завжди.

Далі ми ґрунтовно зупинилися на особливостях застосування програми Maxima до обчислення *R*-інтегралів методом заміни змінної і можемо підвести такі підсумки щодо цього:

1) операція заміни змінної складна для комп'ютерної реалізації і марно сподіватися, що комп'ютер зможе правильно зробити в довільній елементарній функції $f(x)$ заміну незалежної змінної на довільну елементарну функцію $x(t)$, навіть якщо така заміна буде теоретично обґрунтованою;

2) при виконанні цієї операції потрібно якомога сильніше контролювати ситуацію на предмет законності виконуваної заміни та можливих дій комп'ютерної системи у неоднозначних ситуаціях;

3) для цього рекомендується впевнитися, чи існує заданий інтеграл; зобразити графік функції $G(t,x)=0$, яка задає підстановку; виразити явно функції $x(t)$ і

$t(x)$ за допомогою команди solve тощо;

4) вбудована команда changevar досить часто працює некоректно, тому краще користуватися визначеною нами користувацькою командою chv.

Для розширення можливостей Maxima щодо обчислення *R*-інтегралів ми запропонували доповнити її набір команд командою реалізації формули інтегрування частинами для *R*-інтеграла і продемонстрували цю команду у роботі.

Взагалі, на сучасному етапі розвитку СКМ справедливими будуть такі тези: 5) бажано перевіряти результати, отримані в одних програмах, за допомогою інших програм або за допомогою письмових викладок; 6) майже завжди у сумнівних випадках доцільно розглядати отримані за допомогою комп'ютера результати як гіпотези, котрі вимагають перевірки.

Список використаної літератури

1. Семеріков С. О. Maxima 5.13: довідник користувача / За ред. академіка АПН України М.І. Жалдака. – Київ, 2007. – 48 с.
2. Компьютерная математика с Maxima: Руководство для школьников и студентов / *Е. А.* Чичкарёв. – М.: ALT Linux, 2009. – 233 с. (Сайт книги: http://books.altlinux.ru/altlibrary/)
3. http://sourceforge.net/projects/maxima/files/
4. Давидов М. О. Курс математичного аналізу: Підручник: У 3-х ч. Ч. 1. Функції однієї змінної. – К.: ВШ, 1990. – 384 с.